Title stata.com

ciwidth twomeans — Precision analysis for a two-means-difference CI

Description Quick start Menu Syntax

Options Remarks and examples Stored results Methods and formulas

References Also see

Description

ciwidth twomeans computes sample size, CI width, and probability of CI width for a CI for a difference between two means from independent samples. It can compute sample size for a given CI width and probability of CI width. Alternatively, it can compute CI width for a given sample size and probability of CI width. It can also compute probability of CI width for a given sample size and CI width. Also see [PSS-3] ciwidth for PrSS analysis for other CI methods.

For power and sample-size analysis for a two-sample mean test, see [PSS-2] power twomeans.

Quick start

Sample size required for a two-sided 95% CI for the difference between two means to have a width no larger than 12 with a probability of 90%, assuming a common standard deviation of 9 ciwidth twomeans, width(12) probwidth(0.9) sd(9)

Same as above, but for an upper one-sided CI ciwidth twomeans, width(12) probwidth(0.9) sd(9) upper

Sample size required for a two-sided 95% CI for the difference between two means to have a width no larger than 12, with known control- and experimental-group standard deviations of 7 and 10, respectively

```
ciwidth twomeans, width(12) sd1(7) sd2(10) knownsds
```

CI width for a total sample size of 74 with balanced group sizes, given a 90% probability that the CI width will be no larger than the estimated value

```
ciwidth twomeans, n(74) probwidth(0.9) sd(9)
```

Same as above, but for sample sizes of 45 and 30 in groups 1 and 2, respectively ciwidth twomeans, n1(45) n2(30) probwidth(0.9) sd(9)

Probability that the CI width is no larger than 12 for a sample size of 50 ciwidth twomeans, width(12) n(50) sd(9)

Menu

Statistics > Power, precision, and sample size

Syntax

```
Compute sample size
      ciwidth twomeans, width(numlist) probwidth(numlist) | options |
 Compute CI width
      ciwidth twomeans, probwidth(numlist) n(numlist) | options |
 Compute probability of CI width
      ciwidth twomeans, width(numlist) n(numlist) | options |
                                  Description
 options
Main
*level(numlist)
                                  confidence level; default is level (95)
*alpha(numlist)
                                  significance level; default is alpha(0.05)
*probwidth(numlist)
                                  probability of CI width; required to compute sample size
                                    and CI width
* width(numlist)
                                  CI width; required to compute sample size and probability
                                    of CI width
* n (numlist)
                                  total sample size; required to compute CI width and probability
                                    of CI width
*n1(numlist)
                                  sample size of the control group
*n2(numlist)
                                  sample size of the experimental group
*nratio(numlist)
                                  ratio of sample sizes, N2/N1; default is nratio(1), meaning
```

equal group sizes compute(N1 | N2) solve for N1 given N2 or for N2 given N1 allow fractional sample sizes nfractional *sd(numlist) common standard deviation of the control and the experimental groups assuming equal standard deviations in both groups; default is sd(1) * sd1 (numlist) standard deviation of the control group; requires sd2() and knownsds * sd2(numlist) standard deviation of the experimental group; requires sd1() and knownsds knownsds request computation assuming known standard deviations for both groups; default is to assume unknown standard deviations lower lower one-sided CI; default is two-sided CI upper upper one-sided CI; default is two-sided CI synonym for option upper onesided treat number lists in starred options or in command arguments as parallel parallel when multiple values per option or argument are specified (do not enumerate all possible combinations of values)

```
Table
 [no]table[(tablespec)]
                                   suppress table or display results as a table;
                                     see [PSS-3] ciwidth, table
 saving(filename [ , replace ]) save the table data to filename; use replace to overwrite
                                     existing filename
Graph
 graph[(graphopts)]
                                   graph results; see [PSS-3] ciwidth, graph
Iteration
 init(#)
                                   initial value for sample size; default is to use a closed-form
                                     normal approximation
 iterate(#)
                                   maximum number of iterations; default is iterate(500)
 tolerance(#)
                                   parameter tolerance; default is tolerance(1e-12)
 ftolerance(#)
                                   function tolerance; default is ftolerance(1e-12)
  no log
                                   suppress or display iteration log
  no dots
                                   suppress or display iterations as dots
 notitle
                                   suppress the title
```

collect is allowed; see [U] 11.1.10 Prefix commands. notitle does not appear in the dialog box.

where tablespec is

```
column[:label] [column[:label] [...]] [, tableopts]
```

column is one of the columns defined below, and label is a column label (may contain quotes and compound quotes).

column	Description	Symbol
level	confidence level	$100(1-\alpha)$
alpha	significance level	α
N	total number of subjects	N
N1	number of subjects in the control group	N_1
N2	number of subjects in the experimental group	N_2
nratio	ratio of sample sizes, experimental to control	N_2/N_1
Pr_width	probability of CI width	$p_{ m width}$
width	CI width	w
sd	common standard deviation	σ
sd1	control-group standard deviation	σ_1
sd2	experimental-group standard deviation	σ_2
_all	display all supported columns	

Column alpha is shown in the default table in place of column level if alpha() is specified.

Columns nratio, sd, sd1, and sd2 are shown in the default table if the corresponding options are specified.

^{*}Specifying a list of values in at least two starred options, or at least two command arguments, or at least one starred option and one argument results in computations for all possible combinations of the values; see [U] 11.1.8 numlist. Also see the parallel option.

Options

Main

level(), alpha(), probwidth(), width(), n(), n1(), n2(), nratio(), compute(), nfractional; see [PSS-3] ciwidth. probwidth() may not be combined with sd1(), sd2(), and knownsds.

sd(numlist) specifies the common standard deviation of the control and the experimental groups assuming equal standard deviations in both groups. The default is sd(1).

sd1(numlist) specifies the standard deviation of the control group. If you specify sd1(), you must also specify sd2() and knownsds. sd1() may not be combined with probwidth().

sd2(numlist) specifies the standard deviation of the experimental group. If you specify sd2(), you must also specify sd1() and knownsds. sd2() may not be combined with probwidth().

knownsds requests that standard deviations of each group be treated as known in the computation. By default, standard deviations are treated as unknown, and the computation is performed for a Student's t-based CI. If knownsds is specified, the computation is performed for a normal-based CI. knownsds may not be combined with probwidth() and is not allowed when computing the probability of CI width.

lower, upper, onesided, parallel; see [PSS-3] ciwidth.

Table

table, table(), notable; see [PSS-3] ciwidth, table.

saving(); see [PSS-3] ciwidth.

Graph

graph, graph(); see [PSS-3] ciwidth, graph. Also see the *column* table for a list of symbols used by the graphs.

Iteration

init(#) specifies the initial value for the estimated sample size for sample-size determination. The estimated sample size is either the control-group size n_1 or, if compute(N2) is specified, the experimental-group size n_2 . The default is to use a closed-form normal approximation to compute an initial sample size.

iterate(), tolerance(), ftolerance(), log, nolog, dots, nodots; see [PSS-3] ciwidth.

The following option is available with ciwidth twomeans but is not shown in the dialog box: notitle; see [PSS-3] ciwidth.

Remarks and examples

stata.com

Remarks are presented under the following headings:

Introduction
Using ciwidth twomeans
Computing sample size
Computing CI width
Computing probability of CI width

This entry describes the ciwidth twomeans command and the methodology for PrSS analysis for a CI for a difference between two means from independent samples. See [PSS-3] Intro (ciwidth) for a general introduction to PrSS analysis, and see [PSS-3] ciwidth for a general introduction to the ciwidth command. For PSS analysis for hypothesis tests, see [PSS-2] power.

Introduction

The analysis of means is one of the most commonly used approaches in many statistical studies. Many applications lead to the study of two independent means, such as studies comparing the average mileage of foreign and domestic cars, the average SAT scores obtained from two different coaching classes, the average yields of a crop using two different fertilizers, and so on. The two populations of interest are assumed to be independent. We are interested in a CI for the difference $\mu_D = \mu_2 - \mu_1$ between the two means μ_2 and μ_1 . The wider the ranges of the CI are, the less precise it is.

The precision of a CI is commonly measured by its width w or, for a symmetric CI such as the CI for the mean difference, by its half-width d, also known as the margin of error. For example, a two-sided two-means-difference CI is formed as $[\widehat{\mu}_D - d, \widehat{\mu}_D + d]$, where $\widehat{\mu}_D$ is the point estimate of the mean difference. The CI width, the distance between the upper and lower limits, is w=2d; it does not depend on the means difference estimate. The smaller the d or w the more precise the CI.

In PrSS analysis, it is usually of interest to determine the sample size that would be sufficient for a CI to have a prespecified width in a future study. Generally, larger sample sizes lead to more precise CIs. To compute the required sample size, we need to know the expression for w. The expression for w depends on various assumptions.

Similarly to the width of a one-mean CI, the CI width w of a two-means-difference CI depends on sample estimates of standard deviations, in the case of unknown standard deviations. Therefore, wwill vary from one sample to another. To ensure that a CI has the desired width in a future study, this sampling variability of w must be accounted for when computing the required sample size. Kupper and Hafner (1989) introduce what we call the probability of CI width, which specifies the probability of a future CI to have the width of no larger than some prespecified CI width for a given sample size. See Methods and formulas for details.

You can use ciwidth twomeans to perform PrSS analysis for a CI for the difference between two independent means. We discuss the command details in the next section.

Using ciwidth twomeans

ciwidth twomeans computes sample size, CI width, or probability of CI width for a two-meansdifference CI. By default, a two-sided CI is assumed, and the confidence level is set to 95%. You may change the confidence level by specifying the level() option. Alternatively, you can specify the significance level in the alpha() option. You can specify the upper and lower options to request upper and lower one-sided CIs. By default, all computations assume a balanced- or equal-allocation design; see [PSS-4] Unbalanced designs for a description of how to specify an unbalanced design.

To compute total sample size, you must specify the CI width in the width() option and the probability of CI width in the probwidth() option. To compute CI width, you must specify the sample size in the n() option and the probability of CI width in the probwidth() option. You can also compute the probability of CI width given the sample size in n() and CI width in width().

Instead of the total sample size, you can compute one of the group sizes given the other one. To compute the control-group sample size, you must specify the compute (N1) option and the sample size of the experimental group in the n2() option. Likewise, to compute the experimental-group sample size, you must specify the compute (N2) option and the sample size of the control group in the n1() option.

For CIs for means, the CI width does not depend on the mean point estimates, the sample means, so they are not needed in the computations.

By default, all computations are performed for a two-sample CI that assumes equal and unknown standard deviations. By default, the common standard deviation is set to one but may be changed by specifying the sd() option. To specify a known common standard deviation, use the knownsds option. To specify different standard deviations, use the respective sd1() and sd2() options. These options must be specified together and in combination with knownsds; they may not be used in combination with sd(). When sd1() and sd2() are specified, the computations are based on a normal z-based CI. The sd1(), sd2(), and knownsds options may not be combined with probwidth().

Instead of the total sample size n(), you can specify individual group sizes in n1() and n2(), or specify one of the group sizes and nratio() when computing CI width or effect size. Also see *Two samples* in [PSS-4] **Unbalanced designs** for more details.

By default, the computed sample size is rounded up. You can specify the nfractional option to see the corresponding fractional sample size; see *Fractional sample sizes* in [PSS-4] **Unbalanced designs** for an example. The nfractional option is allowed only for sample-size determination.

Some of ciwidth twomeans's computations require iteration. For example, when standard deviations are equal but unknown, the sample-size computation requires iteration. The default initial value of the estimated sample size is obtained by using a closed-form normal approximation. It may be changed by specifying the init() option. See [PSS-3] ciwidth for the descriptions of other options that control the iteration procedure.

In the following sections, we describe the use of ciwidth twomeans accompanied by examples for computing sample size, CI width, and probability of CI width.

Computing sample size

To compute the sample size required for a two-means-difference CI to have the width no larger than a target width, you must specify the target CI width in the width() option and the desired probability of achieving the target CI width in the probwidth() option.

Example 1: Sample size for a two-means-difference CI

Similarly to the study in [PSS-2] **power twomeans**, we want to investigate the effects of smoking on lung function of males. In that entry, we tested the means of the forced expiratory volume (FEV), measured in liters (L), across smokers and nonsmokers, where better lung function implied higher values of FEV. Here, we wish to estimate the CI for the difference between the mean FEV of the two groups.

We are designing a new study for this objective, and we wish to find out how many subjects we need to enroll so that the width of a two-sided 95% CI for the mean FEV difference is no larger than 0.5 L with a probability of 0.96.

Assuming equal numbers of subjects in each group and a common standard deviation of 1, we compute the required sample size:

```
. ciwidth twomeans, probwidth(0.96) width(0.5)
Performing iteration ...
Estimated sample sizes for a two-means-difference CI
Student's t two-sided CI assuming sd1 = sd2 = sd
Study parameters:
        level =
                   95.00
     Pr_width =
                   0.9600
        width =
                   0.5000
           sd =
                   1.0000
Estimated sample sizes:
                      286
  N per group =
                      143
```

We need a total sample of 286 subjects, 143 per group.

The default computation is for the case of equal and unknown standard deviations, as indicated by the output. You can specify the knownsds option to request the computation assuming known standard deviations, but note that knownsds cannot be used in conjunction with probwidth().

4

Example 2: Computing one of the group sizes

Suppose we anticipate a sample of 120 nonsmoking subjects. We wish to compute the required number of subjects in the smoking group, leaving all other study parameters unchanged from example 1. We specify the number of subjects in the nonsmoking group in the n1() option and specify the compute(N2) option.

```
. ciwidth twomeans, probwidth(0.96) width(0.5) n1(120) compute(N2)
Performing iteration ...
Estimated sample sizes for a two-means-difference CI
Student's t two-sided CI assuming sd1 = sd2 = sd
Study parameters:
        level =
                    95.00
     Pr width =
                   0.9600
        width =
                  0.5000
                   1.0000
           N1 =
                      120
Estimated sample sizes:
            N =
                      296
           N2 =
                      176
```

We need a sample of 176 smoking subjects given a sample of 120 nonsmoking subjects.

4

Example 3: Unbalanced design

By default, ciwidth twomeans computes sample size for a balanced- or equal-allocation design. If we know the allocation ratio of subjects between the groups, we can compute the required sample size for an unbalanced design by specifying the nratio() option.

Continuing with example 1, suppose that we anticipate on recruiting twice as many smokers as nonsmokers; that is, $n_2/n_1=2$. We specify the nratio(2) option to compute the required sample size for the specified unbalanced design.

```
. ciwidth twomeans, probwidth(0.96) width(0.5) nratio(2)
Performing iteration ...
Estimated sample sizes for a two-means-difference CI
Student's t two-sided CI assuming sd1 = sd2 = sd
Study parameters:
        level =
                    95.00
     Pr_width =
                   0.9600
        width =
                  0.5000
           sd =
                   1.0000
        N2/N1 =
                   2.0000
Estimated sample sizes:
            N =
                      321
           N1 =
                      107
           N2 =
                      214
```

We need a total sample size of 321 subjects, which is larger than the required total sample size for the corresponding balanced design from example 1.

Also see Two-samples in [PSS-4] Unbalanced designs for more examples of unbalanced designs.

4

Computing CI width

To compute the CI width, you must specify the sample size in the n() option and the desired probability of achieving the target CI width in the probwidth() option.

Example 4: Computing CI width for a two-means-difference CI

Suppose that we have enough resources to enroll 250 subjects in our study on FEV across smokers and nonsmokers. Further suppose that we would like to be 96% certain that the width of a future CI for this sample size will be no larger than the value we estimate. Given these parameters, we compute the CI width as follows:

```
. ciwidth twomeans, probwidth(0.96) n(250)
Estimated width for a two-means-difference CI
Student's t two-sided CI assuming sd1 = sd2 = sd
Study parameters:
        level =
                    95.00
            N =
                      250
  N per group =
                      125
     Pr_width =
                   0.9600
           sd =
                   1.0000
Estimated width:
        width =
                   0.5373
```

The estimated width for a future two-sided 95% CI for the mean difference between the smoking and nonsmoking groups is 0.54, given other parameters.

Computing probability of CI width

To compute the probability that the width of a future CI will be no larger than the specified width, you must specify the sample size in the n() option and the target CI width in the width() option.

Example 5: Computing probability of CI width for a two-means-difference CI

Continuing with example 1, suppose that we have only enough resources to enroll a total of 250 subjects, instead of the 286 we computed before. Assuming equal-sized groups, and given this smaller sample size, we would like to know the probability of obtaining the target CI width of 0.5 for a two-sided 95% CI:

. ciwidth twomeans, width(0.5) n(250) Estimated probability of width for a two-means-difference CI Student's t two-sided CI assuming sd1 = sd2 = sd Study parameters: level = 95.00 N = 250 N per group = 125 width = 0.5000 1.0000 Estimated probability of width:

The estimated probability, given the total sample size of 250, is 54% and is rather low.

1

Example 6: Multiple values of study parameters

0.5427

Pr width =

As a variation of example 5, we would like to see how increasing the sample size, from the 250 we specified above, affects the probability of achieving a target CI width of 0.5. We compute the probability of CI width for a range of sample sizes between 250 and 300, with the step size of 10, by specifying the corresponding numlist in the n() option.

. ciwidth twomeans, width (0.5) n(250(10)300)Estimated probability of width for a two-means-difference CI Student's t two-sided CI assuming sd1 = sd2 = sd

level	N	N1	N2 P	r_width	width	sd
95	250	125	125	.5427	.5	1
95	260	130	130	.7129	.5	1
95	270	135	135	.8467	.5	1
95	280	140	140	.9316	.5	1
95	290	145	145	.9749	.5	1
95	300	150	150	.9925	.5	1

Assuming a balanced design, the probability of CI width increases from 54% to 99% as the sample size increases from 250 to 300, given a target CI width of 0.5.

For multiple values of parameters, the results are automatically displayed in a table, as we see above. For more examples of tables, see [PSS-3] ciwidth, table. If you wish to produce sample-size and other curves, see [PSS-3] ciwidth, graph.

Example 7: One-sided CI

By default, ciwidth twomeans performs computations based on a two-sided CI. You can specify the upper or lower option to request either an upper or lower one-sided CI.

Suppose we want to know the probability of achieving a smaller target CI width of 0.25 for an upper one-sided 95% CI for the difference between FEV means, given a smaller sample size of 200.

```
. ciwidth twomeans, width(0.25) n(200) upper
Estimated probability of width for a two-means-difference CI
Student's t upper CI assuming sd1 = sd2 = sd
Study parameters:
        level =
                    95.00
            N =
                      200
  N per group =
                      100
        width =
                   0.2500
           sd =
                   1.0000
Estimated probability of width:
     Pr width =
                   0.9199
```

Given a total sample size of 200, we are 92% likely to obtain an upper one-sided 95% CI with the width no larger than 0.25 in a future study.

4

Stored results

ciwidth twomeans stores the following in r():

```
Scalars
    r(level)
                       confidence level
    r(alpha)
                       significance level
                       total sample size
    r(N)
    r(N_a)
                       actual sample size
    r(N1)
                       sample size of the control group
    r(N2)
                       sample size of the experimental group
    r(nratio)
                       ratio of sample sizes, N2/N1
    r(nratio_a)
                       actual ratio of sample sizes
    r(nfractional)
                       1 if nfractional is specified, 0 otherwise
    r(onesided)
                       1 for a one-sided CI, 0 otherwise
    r(Pr_width)
                       probability of CI width
    r(Pr_width_a)
                       actual probability of CI width (for sample-size determination when probwidth() specified)
    r(width)
    r(width_a)
                       actual CI width (for sample-size determination when knownsds specified)
                       common standard deviation of the control and experimental groups
    r(sd)
    r(sd1)
                       standard deviation of the control group
    r(sd2)
                       standard deviation of the experimental group
                       1 if option knownsds is specified, 0 otherwise
    r(knownsds)
    r(separator)
                       number of lines between separator lines in the table
    r(divider)
                       1 if divider is requested in the table, 0 otherwise
    r(init)
                       initial value for sample size
    r(maxiter)
                       maximum number of iterations
                       number of iterations performed
    r(iter)
    r(tolerance)
                       requested parameter tolerance
    r(deltax)
                       final parameter tolerance achieved
    r(ftolerance)
                       requested distance of the objective function from zero
    r(function)
                       final distance of the objective function from zero
    r(converged)
                       1 if iteration algorithm converged, 0 otherwise
```

```
Macros
    r(type)
                      Сi
    r(method)
                      twomeans
    r(onesidedci)
                      upper or lower (for a one-sided CI)
    r(columns)
                      displayed table columns
    r(labels)
                      table column labels
    r(widths)
                      table column widths
    r(formats)
                      table column formats
Matrices
    r(pss_table)
                      table of results
```

Methods and formulas

Consider two independent samples with n_1 subjects in the control group and n_2 subjects in the experimental group. Let $\mathbf{x}_1 = (x_{11}, \dots, x_{1n_1})$ be a random sample of size n_1 from a normal population with mean μ_1 and variance σ_1^2 . Let $\mathbf{x}_2 = (x_{21}, \dots, x_{2n_2})$ be a random sample of size n_2 from a normal population with mean μ_2 and variance σ_2^2 . We are interested in a CI for the mean difference $\mu_2 - \mu_1$ estimated using samples \mathbf{x}_1 and \mathbf{x}_2 . Let $\mathbf{x} = (\mathbf{x}_1, \mathbf{x}_2)$. A general two-sided CI is defined as $[ll(\mathbf{x}), ul(\mathbf{x})]$, a lower one-sided CI as $[ll(\mathbf{x}), \infty)$, and an upper one-sided CI as $(-\infty, ul(\mathbf{x})]$, where $ll(\mathbf{x}) = ll$ and $ul(\mathbf{x}) = ul$ are the respective lower and upper confidence limits. Let w be the CI width.

The sample means and variances for the two independent samples are

$$\overline{x}_1 = \frac{1}{n_1} \sum_{i=1}^{n_1} x_{1i}$$
 and $s_1^2 = \frac{1}{n_1 - 1} \sum_{i=1}^{n_1} (x_{1i} - \overline{x}_1)^2$

$$\overline{x}_2 = \frac{1}{n_2} \sum_{i=1}^{n_2} x_{2i}$$
 and $s_2^2 = \frac{1}{n_2 - 1} \sum_{i=1}^{n_2} (x_{2i} - \overline{x}_2)^2$

Let \overline{x}_D be the sample mean difference $\overline{x}_2 - \overline{x}_1$.

A two-sided CI for the means difference is constructed as

$$[\overline{x}_D - w/2, \overline{x}_D + w/2]$$

where w/2 is the half-width or margin of error.

Lower and upper one-sided CIs are

$$[\overline{x}_D - w, \infty)$$
$$(-\infty, \overline{x}_D + w]$$

We use the CI width w as our measure of CI precision. Let $100(1-\alpha)\%$ denote the confidence level, where $0 \le \alpha \le 1$ is the corresponding significance level.

Let $R = n_2/n_1$ denote the allocation ratio. Then $n_2 = R \times n_1$ and CI width can be viewed as a function of n_1 . Therefore, for sample-size determination, the control-group sample size n_1 is computed first. The experimental-group size n_2 is then computed as $R \times n_1$, and the total sample size is computed as $n = n_1 + n_2$. By default, sample sizes are rounded to integer values; see Fractional sample sizes in [PSS-4] Unbalanced designs for details.

PrSS analysis using ciwidth twomeans can be performed under three different assumptions: 1) population standard deviations are known and equal; 2) population standard deviations are known and unequal; and 3) population standard deviations are unknown but equal. We describe each case below.

The following formulas are based on Kupper and Hafner (1989), Ryan (2013), Dixon and Massey (1983), Zar (2010), and Chow et al. (2018).

Methods and formulas are presented under the following headings:

Known equal and unequal standard deviations Unknown and equal standard deviations

Known equal and unequal standard deviations

Below we present formulas for the computations that assume unequal standard deviations. When standard deviations are equal, the corresponding formulas are special cases of the formulas below with $\sigma_1 = \sigma_2 = \sigma$.

Let σ_D denote the standard deviation of the difference between the two sample means. With known standard deviation, $\sigma_D = \sqrt{\sigma_1^2/n_1 + \sigma_2^2/n_2}$. The statistic

$$TS = \frac{\overline{x}_D - (\mu_2 - \mu_1)}{\sigma_D}$$

follows a normal distribution.

Let $z_{1-\alpha}$ be the $(1-\alpha)$ th quantile of a standard normal distribution. Based on the normal distribution of z, the CIs are

$$\begin{cases} \left[\overline{x}_D - z_{1-\alpha/2}\sigma_D, \overline{x}_D + z_{1-\alpha/2}\sigma_D\right] & \text{for a two-sided CI} \\ \left[\overline{x}_D - z_{1-\alpha}\sigma_D, \infty\right) & \text{for a lower CI} \\ \left(-\infty, \overline{x}_D + z_{1-\alpha}\sigma_D\right] & \text{for an upper CI} \end{cases}$$

After expanding σ_D , the corresponding width w is

$$w = \begin{cases} 2z_{1-\alpha/2}\sqrt{\sigma_1^2/n_1 + \sigma_2^2/n_2} & \text{for a two-sided CI} \\ z_{1-\alpha}\sqrt{\sigma_1^2/n_1 + \sigma_2^2/n_2} & \text{for lower and upper one-sided CIs} \end{cases}$$

The control-group sample size n_1 is computed as follows:

$$n_1 = \begin{cases} 4\left(\frac{z_{1-\alpha/2}}{w}\right)^2 \left(\sigma_1^2 + \frac{\sigma_2^2}{R}\right) & \text{for a two-sided CI} \\ \left(\frac{z_{1-\alpha}}{w}\right)^2 \left(\sigma_1^2 + \frac{\sigma_2^2}{R}\right) & \text{for lower and upper one-sided CIs} \end{cases}$$
 (1)

If one of the group sizes is known, the other one is computed using the following formula. For example, for a two-sided CI, to compute n_1 given n_2 , we use the following formula:

$$n_1 = \frac{\sigma_1^2}{\left(\frac{w}{2z_{1-\alpha/2}}\right)^2 - \frac{\sigma_2^2}{n_2}} \tag{2}$$

Unknown and equal standard deviations

When the standard deviations of the control group and the experimental group are unknown and equal, the statistic

$$t = \frac{\overline{x}_D - (\mu_2 - \mu_1)}{s_D}$$

follows a Student's t distribution with ν degrees of freedom. The estimated standard deviation of the sample mean difference s_D is

$$s_D = s_p \sqrt{1/n_1 + 1/n_2}$$

where $s_p = \left\{ \sum_{i=1}^{n_1} (x_{1i} - \overline{x}_1)^2 + \sum_{i=1}^{n_2} (x_{2i} - \overline{x}_2)^2 \right\} / (n_1 + n_2 - 2)$ is the pooled-sample standard deviation

The degrees of freedom ν is

$$\nu = n_1 + n_2 - 2$$

Let $t_{\nu,\alpha}$ denote the α th quantile of a Student's t distribution with ν degrees of freedom. The CIs are

$$\begin{cases} \left[\overline{x}_D - t_{\nu,1-\alpha/2}s_D, \overline{x}_D + t_{\nu,1-\alpha/2}s_D\right] & \text{for a two-sided CI} \\ \left[\overline{x}_D - t_{\nu,1-\alpha}s_D, \infty\right) & \text{for a lower CI} \\ \left(-\infty, \overline{x}_D + t_{\nu,1-\alpha}s_D\right] & \text{for an upper CI} \end{cases}$$

Similarly to the case of an unknown standard deviation for a one-mean CI, the CI width depends on the sample standard deviations. Using the fact that statistic $\nu s_D^2/\sigma_D^2$ follows a χ^2 distribution with ν degrees of freedom, we can compute the probability that the CI width is no larger than a specified value w.

The probability of CI width is

$$\Pr(w) = \begin{cases} \chi_{\nu}^{2} \left\{ \frac{\nu w^{2}}{4t_{\nu,1-\alpha/2}^{2}\sigma^{2}\left(\frac{1}{n_{1}} + \frac{1}{n_{2}}\right)} \right\} & \text{for a two-sided CI} \\ \chi_{\nu}^{2} \left\{ \frac{\nu w^{2}}{t_{\nu,1-\alpha}^{2}\sigma^{2}\left(\frac{1}{n_{1}} + \frac{1}{n_{2}}\right)} \right\} & \text{for lower and upper one-sided CIs} \end{cases}$$

$$(4)$$

where $\chi^2_{\nu}\left(\cdot\right)$ is the c.d.f. of a χ^2 distribution with ν degrees of freedom.

We can compute the desired CI width from (4):

$$w = \begin{cases} 2t_{\nu,1-\alpha/2}\sigma\sqrt{\frac{\chi_{\nu}^{2}\Pr(w)}{\nu}\left(\frac{1}{n_{1}} + \frac{1}{n_{2}}\right)} & \text{for a two-sided CI} \\ t_{\nu,1-\alpha}\sigma\sqrt{\frac{\chi_{\nu}^{2}\Pr(w)}{\nu}\left(\frac{1}{n_{1}} + \frac{1}{n_{2}}\right)} & \text{for lower and upper one-sided CIs} \end{cases}$$
 (5)

where $\chi^2_{\nu,p}$ is the pth quantile of a χ^2 distribution with ν degrees of freedom.

We can solve for the sample sizes iteratively from (5) using initial values obtained from (1) and (2).

References

Chow, S.-C., J. Shao, H. Wang, and Y. Lokhnygina. 2018. Sample Size Calculations in Clinical Research. 3rd ed. Boca Raton, FL: CRC Press.

Dixon, W. J., and F. J. Massey, Jr. 1983. Introduction to Statistical Analysis. 4th ed. New York: McGraw-Hill.

Juul, S., and M. Frydenberg. 2021. An Introduction to Stata for Health Researchers. 5th ed. College Station, TX: Stata Press.

Kupper, L. L., and K. B. Hafner. 1989. How appropriate are popular sample size formulas? *American Statistician* 43: 101–105. https://doi.org/10.2307/2684511.

Ryan, T. P. 2013. Sample Size Determination and Power. Hoboken, NJ: Wiley.

Zar, J. H. 2010. Biostatistical Analysis. 5th ed. Upper Saddle River, NJ: Pearson.

Also see

- [PSS-3] **ciwidth** Precision and sample-size analysis for CIs
- [PSS-3] ciwidth, graph Graph results from the ciwidth command
- [PSS-3] ciwidth, table Produce table of results from the ciwidth command
- [PSS-2] power twomeans Power analysis for a two-sample means test
- [PSS-5] Glossary
- [R] **ttest** t tests (mean-comparison tests)
- [R] **ztest** z tests (mean-comparison tests, known variance)

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on citing Stata documentation.